# $Id: PKGBUILD 181990 2013-04-05 08:56:58Z svenstaro $
# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=kdemultimedia-mplayerthumbs
pkgver=4.10.2
pkgrel=1
pkgdesc="MPlayer based thumbnail generator for video files"
url='https://projects.kde.org/projects/kde/kdemultimedia/mplayerthumbs'
arch=('i686' 'x86_64')
groups=('kde' 'kdemultimedia')
license=('GPL' 'LGPL' 'FDL')
depends=('kdebase-runtime' 'mplayer')
makedepends=('cmake' 'automoc4')
source=("http://download.kde.org/stable/${pkgver}/src/mplayerthumbs-${pkgver}.tar.xz")
sha1sums=('80f35eb2e30382984cb3529c4226f911f5e2c30a')

build() {
  cd "${srcdir}"
  mkdir build
  cd build
  cmake ../mplayerthumbs-${pkgver} \
    -DCMAKE_BUILD_TYPE=Release \
    -DKDE4_BUILD_TESTS=OFF \
    -DCMAKE_INSTALL_PREFIX=/usr
  make
}

package() {
  cd "${srcdir}"/build
  make DESTDIR="${pkgdir}" install
}
